
CORSO DI LAUREA SPECIALISTICA IN INGEGNERIA MECCANICA 
CORSO DI PROGETTAZIONE ASSISTITA DELLE STRUTTURE MECCANICHE II 

 
ANALISI DEL CONTATTO SFERA-PIANO 

 
Si analizzi il contatto tra una sfera ed un piano semi-infinito, come da figura 1, determinando: 

• l'ampiezza dell'area di contatto 
• l'andamento della pressione di contatto 
• lo stato di tensione nei due corpi in contatto 

Dati del problema: 
• Diametro sfera  (D) = 10 mm 
• Carico massimo applicato (P) =  400 N 
• Modulo elastico materiale sfera e piano (E) = 206000 MPa 
• Coefficiente di Poisson dei due materiali (ν) = 0.3 

Secondo la teoria di Hertz, la dimensione dell'area di contatto nel caso specifico è data da: 

3
/2

908.0
D

Pa Δ
=  

dove: 

⎟⎟
⎠

⎞
⎜⎜
⎝

⎛ −
=Δ

E

212 ν
 

mentre il valore della massima pressione di contatto è dato da: 

L'andamento normalizzato delle principali componenti di tensione agenti nella lastra piana (sul prolungamento dell’asse 
verticale della sfera)  in funzione della distanza dalla superficie è riportato nella Fig. 2. 
Un modello suggerito per l'analisi del problema è mostrato nella Fig. 3. Esso fa uso di elementi "gap" del tipo “surface-
to-surface”. 
Il file di ingresso per l’ottenimento del modello è riportato in Appendice. In tale file, notare in particolare: 

• L’impiego del comando ESURF per la generazione automatica degli elementi di contatto 
• L’impiego dei comandi PATH per la creazione di un grafico delle tensioni in funzione della profondità 
• L’impiego delle variabili dimensionate (*DIM) per la creazione di grafici in cui si confrontano previsioni 

teoriche e valori calcolati (*VPLOT) 
• L’uso del comando MULTIPRO per la creazione di finestre di ingresso 
• L’impiego dei comandi LESIZE e KEREFINE per ottenere una mesh opportunamente infittita nella zona di 

contatto 
Avvertenza: per visualizzare correttamente la zona di contatto è necessario far rappresentare gli spostamenti in vera 
grandezza (Plo Controls   ---- > Displacement Scaling) 
Ulteriori analisi: utilizzare per i materiali un modello di comportamento elasto-plastico. 



 
 

 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 

 
 

Fig. 3 – Mesh suggerito 
 
 
 

 
Fig. 2 (Asse Z = asse sfera) 

Fig. 1 



APPENDICE : FILE DI INGRESSO 
 
C*** 
C*** ANALISI DI CONTATTO HERTZIANO SFERA PIANO 
C***  
FINISH 
/CLEAR 
/PREP7 
 
MULTIPRO,'start',9 
*CSET,1,3,D,'Diametro sfera [m] ',0.01 
*CSET,4,6,RP1,'Altezza piano/diametro ',1 
*CSET,7,9,RP2,'Larghezza piano/diametro ',1 
*CSET,10,12,NR,'N. elementi su raggio sfera ',15 
*CSET,13,15,SP1,'Fatt.incr. elem. da zona cont. ',20 
*CSET,16,18,FR1,'Fattore infitt. zona cont. ',1 
*CSET,19,21,DR1,'N. elem. per infitt.',10 
*CSET,22,24,PMAX,'Carico massimo [N] ',400 
*CSET,25,27,NSTP,'N. load step ',10 
MULTIPRO,'end' 
 
MULTIPRO,'start',4 
*CSET,1,3,ES,'Modulo Young sfera [Pa] ',206000e6 
*CSET,4,6,NIS,'Coeff. di Poisson sfera ',0.3 
*CSET,7,9,EP,'Modulo Young piano [Pa] ',206000e6 
*CSET,10,12,NIP,'Coeff. di Poisson piano',0.3 
MULTIPRO,'end' 
 
PI=3.14157 
 
H=RP1*D         !Altezza del piano nel modello 
B=RP2*D  ! Larghezza del piano nel modello 
        ! NR= Numero di elementi nella sfera in senso radiale 
 ! SP1 = fATTORE DI SPAZIATURA PER LA DIMENSIONE DEGLI ELEMENTI A PARTIRE DAL 
PUNTO DI CONTATTO 
NC=NR*PI/2    !Numero di elementi nella sfera sulla superficie esterna 
NH=NR*RP1*2  !!Numero di elementi verticali nel piano 
NB=NR*RP2*2               !Numero di elementi ORIZZONTALI nel piano 
 !FR1 = fattore di riduzione della dimensione degli elementi all'apice della zona di contatto 
 !DR1 = N° di elementi nella zona di contatto influenzati dalla revisione 
 !PMAX = Carico massimo applicato 
 !NSTP = n. step di carico 
 
C*** variabili dimensionate per output 
C*** 
*DIM,RISU,TABLE,NSTP,5 
 
*DO,IJK,1,NSTP 
 RISU(IJK,0)=IJK 
*ENDDO 
*DO,IJK,1,5 
 RISU(0,IJK)=IJK 
*ENDDO 
 
 
C***  MATERIALE SFERA 
MP,EX,1,ES  
mp,nuxy,1,NIS 
 
 
C***  MATERIALE PIANO 



MP,EX,2,EP  
mp,nuxy,2,NIP 
 
 
C*** TIPI DI ELEMENTO 
ET,1,42,,,1 ! Elemento assialsimmetrico 
 
ET,2,169 ! Target element 
ET,3,171,,,,,3 ! Contact element 
C***ET,3,175,,,,,3 ! Contact element 
 
 
C*** COSTRUZIONE MODELLO  
 
wplane,,0,d/2, 
 
PCIRC,0,D/2,-90,0  ! Sfera 
RECTNG,0,B,-D/2,-(D/2+H) ! Piano 
 
 
C*** IMPOSTAZIONE SUDDIVISIONE LINEE 
 
asel,,area,,1,1    ! Arco della sfera 
lsla 
lsel,r,loc,x,D/2/1000,D/2*0.999 
LSEL,R,LOC,Y,D/2/1000,D/2*0.999 
*GET,LNN,LINE,0,NUM,MAX 
*GET,KP1,LINE,LNN,KP,1 
XP1=KX(KP1) 
*IF,XP1,LT,D/2/1000,THEN 
 LESIZE,ALL,,90/NC,,SP1 
*ELSE 
 LESIZE,ALL,,90/NC,,1/SP1 
*ENDIF 
 
lsel,,loc,x,-1,D/2/1000   ! Linea verticale sfera 
lsel,r,loc,y,0,d/2 
*GET,LNN,LINE,0,NUM,MAX 
*GET,KP1,LINE,LNN,KP,1 
YP1=KY(KP1) 
*IF,YP1,LT,D/2/1000,THEN 
 lesize,all,,,nr,SP1 
*ELSE 
 lesize,all,,,nr,1/SP1 
*ENDIF 
 
lsel,,loc,y,d/2*0.999,d   ! Linea orizzontale sfera 
lesize,all,,,nr 
 
lsel,,loc,y,-d/2/1000,d/2/1000  ! Linea orizzontaloe superiore piano 
*GET,LNN,LINE,0,NUM,MAX 
*GET,KP1,LINE,LNN,KP,1 
XP1=KX(KP1) 
*IF,XP1,LT,D/2/1000,THEN 
 lesize,all,,,nB,SP1 
*ELSE 
 lesize,all,,,nB,1/SP1 
*ENDIF 
 
 
lsel,,loc,x,-1,D/2/1000   ! Linea verticale sinistra piano 
lsel,r,loc,y,-H,0 



*GET,LNN,LINE,0,NUM,MAX 
*GET,KP1,LINE,LNN,KP,1 
YP1=KY(KP1) 
*IF,YP1,LT,D/2/1000,THEN 
 lesize,all,,,nH,SP1 
*ELSE 
 lesize,all,,,nH,1/SP1 
*ENDIF 
 
c***ksel,,loc,x,-1,d/2/1000 
C***ksel,r,loc,y,-d/2/1000,d/2/1000 
C***kesize,,d/2/nr/FR1 
 
 
C*** MESH 
C*** 
 
alls 
mshkey,2 ! mapped, se possibile 
mshape,0 ! elementi quadrilateri 
 
TYPE,1  ! MESH SFERA 
MAT,1 
AMESH,1 
 
MAT,2  ! MESH PIANO 
AMESH,2 
 
 
C*** RAFFINAMENTO MESH NELLA ZONA DI CONTATTO 
 
ksel,,loc,x,-1,d/2/1000 
ksel,r,loc,y,-d/2/1000,d/2/1000 
KREFINE,ALL,,,fr1,dr1,clean 
 
 
C*** CREAZIONE ELEMENTI DI CONTATTO 
 
asel,,area,,1,1    ! SFERA (TARGET) 
lsla 
lsel,r,loc,x,D/2/1000,D/2*0.999 
LSEL,R,LOC,Y,D/2/1000,D/2*0.999 
NSLL 
TYPE,2 
ESURF 
 
lsel,,loc,y,-d/2/1000,d/2/1000  ! PIANO (CONTACT) 
NSLL 
TYPE,3 
ESURF 
 
ALLS 
 
Nsel,,loc,x,-1,d/2/1000 
Nsel,r,loc,y,-d/2/1000,d/2/1000 
CP,NEXT,UY,ALL 
 
C*** SOLUZIONE 
/SOLU 
  
C*** VINCOLI E CARICHI 
NSEL,Y,-H*1.001,-H*0.999       !Vincolo sotto il piano 



D,ALL,UY,0 
NALL  
 
 
DELT = 2*(1-NIP**2)/EP 
 
*DO,IJK,1,NSTP 
 TIME,IJK 
 NSEL,,LOC,Y,D/2*0.999,D/2*1.001 
 CDSF=PMAX/(PI*D**2/4)*IJK/NSTP 
 SF,ALL,PRESS,CDSF       !Carico imposto  
 RISU(IJK,1)=CDSF 
 AAA=0.908*(PMAX*IJK/NSTP*DELT*D/2)**(1/3) 
 P0=0.578*(PMAX*IJK/NSTP/D/D*4/DELT/DELT)**(1/3) 
 RISU(IJK,2)=AAA 
 RISU(IJK,4)=P0 
 ALLS 
 
 NSUBST,30               !N° substep 
 AUTOTS,ON               !Ottimizzazione substep 
 KBC,0                  !Variazione lineare del carico nei substep 
      
 OUTRES,ALL,ALL     
 NEQIT,30                !N° massimo di iterazioni 
 
 LSWRITE,IJK                
*ENDDO 
 
LSSOLVE,1,NSTP 
FINISH  
 
/POST1 
SET,LAST 
 
C*** Grafico delle tensioni in funzione della profondità 
PATH,PIPPO,2,,100 
PPATH,1,,0,0 
PPATH,2,,0,-H*0.1 
PDEF,SY,S,Y 
PDEF,SX,S,X 
PLPATH,SY,SX 
 
*ASK,IFLAG,RETURN PER CONTINUARE 
 
C*** LETTURA DELLE DIMENSIONI DELLA ZONA DI CONTATTO CALCOLATA 
C*** E DELLA RELATIVA PRESSIONE MASSIMA 
C*** 
*DO,IJK,1,NSTP 
 SET,IJK   ! Pressione massima 
 ASEL,,AREA,,2,2 
 ESLA 
 NSLE 
 NP1=NODE(0,0,0) 
 *GET,PMX,NODE,NP1,S,Y 
 RISU(IJK,5)=-PMX 
 alls 
 
 ESEL,,TYPE,,3,3  ! dimensione area di contatto 
 NSLE 
 NSEL,R,CONT,PRES,-PMX/1000,-PMX*2 
        *GET,CLDIST,NODE,0,MXLOC,X 
 RISU(IJK,3)=CLDIST 



 ALLS 
 
*ENDDO 
 
C*** VISUALIZZAZIONE GRAFICI DELLE GRANDEZZE CALCOLATE E TEORICHE 
C*** IN FUNZIONE DEL CARICO 
 
/AUTO 
/VIEW,,0,0,1 
/GCOLUMN,1,THEORETICAL 
/GCOLUMN,2,CALCULATED 
/AXLAB,X,LOAD [N] 
 
C*** Area di contatto 
/AXLAB,Y,CONTACT RADIUS [mm] 
*VPLOT,RISU(1,1),RISU(1,2),3 
/REPLOT 
*ASK,DUMMY,Return per continuare 
 
C*** Pressione massima 
/AXLAB,Y,CONTACT PRESSURE [Pa] 
*VPLOT,RISU(1,1),RISU(1,4),5 
/REPLOT 
 
 
 
 

 


